
Programming the Rev Language

Introduction

The Rev programming language was originally created as a project to implement Peter Grogono’s 
Mouse language, seen by me originally in the July, 1979, issue of Byte magazine.  While implementing
the base concepts of Mouse, it seemed to me that by refactoring the language and changing the order of
some of the operators, the language could become more powerful and add additional features while 
retaining the original look and feel of the original language.

According to Peter Grogon, the name for Mouse came about because it described something small and 
active.  In the spirit of the name, Rev has a few reasons for the name: 1. Räv is Swedish for fox, 
another small, active animal.  2. Revving an engine is something that’s fun and cool to do.  3. 
Coincidentally, my mainframe username at university was Revjim.

So What is Rev?

Rev is an interpreted language where each character is an operation, a variable, or whitespace.  It is 
somewhat experimental, in that the goal is to do all sorts of things that modern languages can do, while 
having a minimal number of commands and operators while, hopefully, attempting to be readable.

The Shortest Program

The shortest possible Rev program is an empty file!  The Rev interpreter will read it in, run it, and 
produce no output.  There are no required start of program statements or end of program statements.  

~ Blank program

The program above is not quite the shortest program, but it does have a comment.  In Rev programs, if 
a tilde (~) is encountered, Rev will skip to the end of the line and continue the code execution at the 
start of the next line.  Each comment is applicable only to an individual line, and there is no way to 
mark a block of comments.

At this point, it’s worth nothing that spacing, code indentation, and whitespace is not important to how 
Rev processes the program, except for separating numbers, which we will soon see.

The Shortest Useful Program (Hello, World)

It is customary when introducing new programming languages to print “Hello, World!” to the screen.  
This is slightly more difficult than most languages, so we’ll start by printing “Hello, World”  without an
exclamation point.

~ Hello World program

"Hello, World!"

The output of this program is:

Hello, World

In Rev, the exclamation point is interpreted as a newline character.  There are ways to print to the 
exclamation point, but we’ll get to those later.

So what is happening in this program?  Rev sees the first quotation mark, and then starts printing 
characters out to the screen.  When it reaches the exclamation point, Rev moves the cursor to the next 
line.  When it reaches the final quotation mark, Rev stops printing and resumes processing new 
commands.

Mathematics

Unlike many programming languages (and like many others), Rev uses postfix notation, where the 
mathematical operator follows the operands.  For example,, 4+3 is represented as “4 3 +”.  Because 
Rev needs to differentiate between where 4 ends and 3 starts (rather than interpreting the number as 
43), it is important to separate them with a space.  This is the only case where whitespace would be 
required in a Rev program.

Rev uses a stack.  You can think of a stack like a stack of books.  You can’t get to the bottom one 
without removing all the ones on top first.  When you write “4 3 +”, what you are actually saying is, 
perform the following steps:

• Put the number four on the stack.  (We call this pushing a value onto the stack)
• Put the number three on the stack.  
• Then, take the two numbers that are on the top of the stack, add them.  We call the process of 

taking something off of the stack “popping”.  In stack terms, we would say, “pop two values off
the stack”.

• Put (push) the result of the addition on to the stack.

Rev has five arithmetic operators: + (addition), - (subtraction), * (multiplication), / (division), and % 
(modulus, or remainder).  All numbers in Rev are integers.  Sorry, no floating point numbers are 
available!  We use the period (.) for a different operation, as we will see later.

To demonstrate these operators, we’ll need one more command: ! (exclamation point).  This command 
takes the stack, pops (removes) the top value from it, and prints it out.  Also note that the order that the 
numbers are pushed onto the stack can be important.  “8 4 /” is not the same as “4 8 /”.

"3+4=" 3 4 + ! "!"
"8/4=" 8 4 / ! "!"
"8-4=" 8 4 - ! "!"
"2*4=" 2 4 * ! "!"
"2%4=" 2 4 % ! "!"

This program outputs:

3+4=7
8/4=2
8-4=4
2*4=8
2%4=2

Notice the use of the “!” set of commands to print out a new line.  If we didn’t do this, everything 
would print out on one long line.  Let’s try it!

"3+4=" 3 4 + ! 
"8/4=" 8 4 / !
"8-4=" 8 4 - !
"2*4=" 2 4 * !
"2%4=" 2 4 % !

The output:

3+4=78/4=28-4=42*4=82%4=2

Not very readable, is it?

There are three logical operators which behave similarly to the arithmetic operators: < (less than), > 
(greater than), and = (equal to).  They work like the five arithmetic operators, except that the push a 
non-zero value (usually one) to the stack if the result is true, and zero if the result is false.

~ Print the results of some math

"3+4=" 3 4 + ! "!"
"8/4=" 8 4 / ! "!"
"8-4=" 8 4 - ! "!"
"2*4=" 2 4 * ! "!"
"2%4=" 2 4 % ! "!"
"2>4=" 2 4 > ! "!"
"2<4=" 2 4 < ! "!"
"2=4=" 2 4 = ! "!"
"4=4=" 4 4 = ! "!"

The output:

3+4=7
8/4=2
8-4=4
2*4=8
2%4=2
2>4=0
2<4=1
2=4=0
4=4=1

Variables

There are 26 named variables in Rev: a, b, c, and the rest up to z.  All variables are lower case.  To 
assign a value to a variable, type the value that you want to assign to a variable, the variable name, and 
colon (:), the assignment operator.  For example, to assign 5 to a, write “5 a:”.  To get the value of a 
variable, we must use the dereference operator, the period (.), which converts the name of a variable to 
the contents of that variable. 

~Variable assignment program

5 a:     ~ Set a to 5
a.!      ~ Print the value of a
"!"      ~ Print a new line

The output:

5

This may seem a bit strange, but what this program says is the following: 
• Put 5 on the stack.  
• Put the name ‘a’ on the stack.
• Take the named variable on top of the stack (a), and assign the value under it (5) to it.
• On the next line, put the name ‘a’ on the stack.
• Then, convert the named variable “a” to its value (5) and put it on the stack.  
• Print out the value on top of the stack.
• Print a new line.

Here’s a slightly more complex version.

~ Another more complex example...adding variables

5 a:   ~ Assign 5 to a
6 b:   ~ Assign 6 to b

a. b. + c:  ~ Add the values of a and b and put it in c

"c=" c.! "!"  ~ Print the value of c

And the results:

c=11

You may be wondering why we jump through all these hoops of dereferencing using the “.” operator 
rather than just saying that when you type “a” that you always mean “the value of a”.  Well, we can do 
all sorts of neat things with the language that we’ll get to soon enough.  Rest assured, this additional set
of steps is necessary to build more complex data structures and write more interesting programs.

Conditional Statements

Like most programming languages, Rev has a conditional pair of operators, [ (left bracket) and ] (right 
bracket).  When Rev reaches [ (left bracket), it pops (removes) the top value off of the stack.  If it is not
zero, it keeps going, executing everything between the brackets.  If it is zero, it skips to the 
corresponding right bracket.

~ Conditional Statement example

5 6 < [ "We should print this!" ]

5 6 > [ "We should not print this!" ]

Output:

We should print this

We can also nest conditional statements.  Consider this more complex version of the same program:

5 6 < [ "We should print this!"
    9 9 = [ "Here is more stuff to be printed!" ]
    9 9 > [ "This shouldn't be printed!" ]
      ]
5 6 > [ "We should not print this!"
    9 9 = [ "This isn't printed!" ]
    9 9 > [ "And neither is this.!" ]
      ]

Output:

We should print this
Here is more stuff to be printed

Note that none of the second if statement (5 6 >) is printed.  If the first condition is not met, none of it 
is met.

It’s also important to note that you not have to use the logical operators with the conditional statement.  
For example:

~ Conditional Statement example

123 [ "We should print this!" ]
0 [ "We should not print this!" ]

And the output:

We should print this

Loops

6



Programming the Rev Language September 3, 2017
James Ghofulpo

Rev has one set of looping commands, “(“ (open parenthesis), and “)” (close parenthesis).  When Rev 
reaches these commands, it will continuously repeat the commands between the parenthesis forever and
ever.  The program:

 ( “x” )

will print “x” over and over again, until the user stops the program manually (usually, by pressing 
control-C).

To get out of a loop, the break operator, ^, must be used.  Like the conditional operators, the break 
operator pops the top value off of the stack.  If it is zero, the loop continues.  If it is not zero, the 
program continues executing the program after the end of the current loop.  

5 a:      ~ Set a to 5
(         ~ Start looping
    a.!   ~ Print the value of a
    "!"   ~ Print a new line
    a.    ~ Put the value of a on the stack
    1     ~ Put one on the stack
    =     ~ If a=1, put 1 on the stack.  Else, 0
    ^     ~ If it is one, exit the loop
    a.    ~ Put the value of a on the stack
    1     ~ Put one on the stack
    -     ~ Put the value of a-1 on the stack
    a:    ~ Set a to the top of the stack
)         ~ End of loop

"Done!"   ~ Ending message

And the output:

5
4
3
2
1
Done

The program is much more elaborately documented than necessary.  A more common, but still 
readable, version of the same program would be:

5 a:        ~ Set a to 5

( a.! "!"   ~ Print the value of a
  a. 1 = ^  ~ Exit if a=1
  a. 1 - a: ~ Decrement a
)           ~ End of loop

"Done!"   ~ Ending message

And the same output:

5
4
3
2
1
Done

Loops, like conditionals, can be nested, or put inside each other.  

~ Example of nested loops

0 a:                 ~ Set a to zero
(
  "a=" a.! " "       ~ Print the value of a and a space
    1 b:             ~ Set b to 1
    (
        "b=" b.! " " ~ Print the value of b
        b.1+b:       ~ Increment b (b = b + 1)
        b. 3 > ^     ~ Exit if b > 3
    )
    "!"              ~ Print a new line
  a. 1 + a:          ~ Increment a
  a. 4 > ^           ~ Exit if a > 4
)                    ~ End of loop

And the output:

a=0 b=1 b=2 b=3 
a=1 b=1 b=2 b=3 
a=2 b=1 b=2 b=3 
a=3 b=1 b=2 b=3 
a=4 b=1 b=2 b=3 

Input

Any computer program is much more useful if you can run it with different sets of inputs.  The input 
operator, ? (question mark), gets an integer from the user.  For example:

"Input value: " ~ Command prompt
? a:            ~ Get a value, and save it to a
( a.! "!"       ~ Print the value of a
  a. 1 = ^      ~ Exit if a=1
  a. 1 - a:     ~ Decrement a
)               ~ End of loop

"Done!"         ~ Ending message

And the output:

Input value: 3
3
2
1
Done

Character Processing

There are two variations of the print and get input commands that can be used for character 
processing: ?’ (question mark, followed immediately by an apostrophe) and !’ (exclamation point, 
followed by an apostrophe).  Like ?, ?’ gets a character of input from the user.  !’ prints a character of 
input from the user.

"Type a character:" ?'
"You typed:" !' "!"

The output:

Type a character:z
You typed:z

Functions

Functions in Rev are delimited by { (open brace) and } (close brace).  The name of the function is a 
single character, specified before the open brace.  For example, a function that prints a new line named 
‘p’ would look like:

p { “!” }

To call the function, first specify the function name, and follow it by a hash mark (#).  To call the 
previous function, write:

p#

There are some caveats about using functions.  First, the function must be declared before it is called.  
This is because, in the above example, p is really just another variable.  When Rev reaches the open 
brace, it is putting the address of the function to run in the variable p.  This also means that you cannot 
reuse the same variable name for a “normal” variable and a “function” variable.

An example of functions would be:

a { "We're in function a!" }
b { "We're in function b!" }

a#   ~ Call a
b#   ~ Call b

And the output is:

We're in function a
We're in function b

The return operator (@, at sign) can be used to exit a function early.  For example:

a { "We're in function a!"
 @  ~ Return early
 "This is not printed!" }

a#   ~ Call a

And the output:

We're in function a

A common use of @ would be used with the conditional operators to exit a function early.

Because of the variable nature of functions, function functionality can be changed dynamically.  

~ Example of a dynamic function

b { "We're in b!" }

b#

b { "We're now in the new version of b!" }

b#

And the output:

We're in b
We're now in the new version of b

When we get a bit further along, we’ll show how this feature allows the user to create things such as 
arrays of functions, structures that contain functions, and other useful data types.

Passing Parameters to Functions

Parameters are passed to functions using the stack.  Values pushed onto the stack are popped off in the 
opposite order:

a { c: d:   ~ Assign the two values on top of the
           ~ stack to 'c' and 'd'
 "c=" c.! "!" ~ Print c
 "d=" d.! "!" ~ Print d
}

10 11 a#

And the output:

c=11
d=10

Notice how c and d are opposite of what you would normally expect if you are used to passing lists of 
variables to a function, due to the use of the stack.

Return values also use the stack.  Because of this, a function can have more than one return value.  For 
example:

a { c: d:    ~ Assign the two values on top of the
            ~ stack to 'c' and 'd'
   c. d. +  ~ Push c+d
   c. d. *  ~ Push c*d
}

10 11 a#

"Results: " ! " " ! "!"

And the results:

Results: 110 21

Note how the multiplication was printed first.  This is because it was pushed to the stack last.

Variables in Functions

In many computer languages, when you enter a function, you get a completely new variable scope.  
That is, variable ‘a’ in a function is often not the same as variable ‘a’ in the function that called it.  For 
example:

~ Variables in functions

b { 5 a:    ~ Assign 5 to a
    "In b, a=" a.! "!" ~ Print it
  }

6 a: ~ Assign 6 to a
b#   ~ Run function b
"In the main function, a=" a.! "!" ~ Print it

Output:

In b, a=5
In the main function, a=6

When we called function b, we kept the original value of a in the main function around, so when we 
exited function b, we got it back again.

If we don’t assign the value of a in function b, Rev will go back through the calling functions and 
return the first value of ‘a’ it finds.  For example:

~ Variables in functions, not assigning a

b {
    "In b, a=" a.! "!" ~ Print it
  }

6 a: ~ Assign 6 to a
b#   ~ Run function b
"In the main function, a=" a.! "!" ~ Print it

Output:

In b, a=6
In the main function, a=6

This can give you a way to have read-only global variables.  But due to this, it becomes easy to confuse
yourself by changing the scope of the variables to the current function, dynamically.  In other words, 
when you are in a function, the value of a variable is whatever it was before you called the function, 
until you set it.  At that point, the value is whatever you set it to, but the value in the calling function is 
retained.

~ Variables in functions

b {
    "In b (before assigning a), a=" a.! "!" ~ Print it
    5 a:    ~ Assign 5 to a
    "In b, a=" a.! "!" ~ Print it
  }

6 a: ~ Assign 6 to a
b#   ~ Run function b
"In the main function, a=" a.! "!" ~ Print it

Output:

In b (before assigning a), a=6
In b, a=5
In the main function, a=6

Ending the Program Early

Sometimes, we just don’t want to run the program anymore.  We can stop execution of the program 
with the exit operator, $ (dollar sign).

"Hello world!"
$  ~ Stop running the program
"This is not printed!"

And the output:

Hello world

Taking a Break

At this point, we have covered the entire language, except for one command, _ (memory allocation)!   
We’ll get to that command a bit later, when we decide to get into more trouble.  First,  it is very 
important that we start discussing some the nuances of Rev that give it a lot of extra power.  But, as a 
wise person once said, with great power comes great responsibility.

Variables and Names

Remember way back when, a few pages ago, when we discussed variables?  

5 a:
a.! "!"

And the corresponding output:

5

What would happen if we forgot the dereference operator (period) when printing out the value of a?

5 a:
a! "!"

The output:

0

What happened?  On the second line, we took the name of a (0, where a=0, b=1, etc.) and put it on the 
stack, rather than the value of a.

Due to this behavior, with a bit of clever math, we can do the following:

5 a:   ~ Set a to 5
6 b:   ~ Set b to 6
a      ~ Put the name 'a' on the stack
1 +    ~ Add one to it
.      ~ Get the value of it
! "!"  ~ And print it

And the output:

6

If you’re familiar with programming languages, it’s important to note the when we type ‘a’, we are not 
putting the address of ‘a’ on the stack, but merely the name ‘a’.  This is not exactly the same as C style 
pointers, but it’s a close analogy.  The main difference is that in Rev, the actual address is evaluated as 
late as possible, allowing the user to increment the name of the variable, rather than just the pointer.

With this same type of math, we can create arrays and do operations on them:

5 a:    ~ Treat 'a' as an array of three elements
6 a1+:  
7 a2+:  

a. a1+. a2+.  ~ Put the values of all three elements
              ~ on the stack
+ +           ~ Add them all together

! "!"         ~ Print the result

And the result:

18

Note that in the above example, it’s important to not use the variables b and c, since a1+ is the same as 
b, and a2+ is the same as c.

Because of the reference by name functionality, we can pass the name of a variable to a function, and 
the function can work on the value directly, much like passing the address of a variable would.

~ Pointers to variables in functions

b { a:  ~ Save off the parameter on the stack to 'a'
    "In b (before assigning a), a=" a..! "!" ~ Print it
    5 a.:    ~ Assign 5 to the memory pointed to by a
    "In b, a=" a..! "!" ~ Print it
  }

6 a: ~ Assign 6 to a
"In the main function, a=" a.! "!" ~ Print it
a b#   ~ Run function b, passing the variable 'a' to the 
function
"In the main function, after calling b, a=" a.! "!" ~ Print it

Output:
In the main function, a=6
In b (before assigning a), a=6
In b, a=5
In the main function, after calling b, a=5

Note that in the function, we are using “a.” instead of “a”.  This is because we are operating on the 
variable ‘a’ that is in the main program, not the variable ‘a’ that is in the function.  That is, the variable 
‘a’ in the function points to the variable ‘a’ in the main function.

Early on, it was mentioned that there were 26 named variables, a-z.  This does not mean that in a given 
scope that there are only 26 variables.  This just means that only the first 26 variables have names.  
Additional variables can be referened by an offset.

~ Anonymous variables

3 1000:    ~ Set variable number 1000 to 3
1000.! "!" ~ Print the value of variable 1000

And the output:

3

Obviously, this can be extremely confusing to understand what’s going on, especially after not having 
looked at the program for some time.  But, just because this practice isn’t recommended doesn’t mean 
you can’t use it.

Nesting Functions

Because of the variable nature of functions, we can also create functions inside of functions that are 
only accessible by that function.  For example:

~ Functions in functions

a {
    b { "This is function b, inside of function a!" }
    "In function a!"
    b#  ~ Call function b
}

b { "This is function b, but not inside anything else!" }

b#  ~ Call function b
a#  ~ Call function a
b#  ~ Call function b again

The output:

This is function b, but not inside anything else
In function a
This is function b, inside of function a
This is function b, but not inside anything else

Note how that calling function b again in the main program does not call function b inside of function 
a.  This is because the function b in function a is only callable from within function a, since ‘b’ is a 

local variable within function ‘a’.  When we exit function a, this value of ‘b’ goes away and the 
meaning of ‘b’ reverts to our original definition.

Since variable ‘b’ is still just that, a variable, we can now do interesting things from functions, like 
returning the function itself to the main program.  

~ Functions in functions

a {
    b { "This is function b, inside of function a!" }
    b. ~ Return the address of this function to the main 
program
}

a#  ~ Call function a
#   ~ Call the function that was returned, directly!

And the output:

This is function b, inside of function a

You may be thinking, is the last line of the program a typo?  How can you have a # command by itself?
Remember, # operates on the last value on the stack (the return value, “b.”).  If this seems confusing, 
you can always set a local variable to do the same thing:

~ Functions in functions, with return value

a {
    b { "This is function b, inside of function a!" }
    b. ~ Return the address of this function to the main 
program
}

a#  ~ Call function a
c:  ~ Assign c to the address returned by function a
c.#   ~ Call the function that was returned, directly!

Output:

This is function b, inside of function a

We can further complicate things by returning more than one function from the called function:

~ Functions in functions, with return value

a {
    a { "This is function a, inside of function a!" }
    b { "This is function b, inside of function a!" }
    a. b.  ~ Return the two functions
}

a#  ~ Call function a
c:  ~ Assign c to the address returned for function b
d:  ~ Assign c to the address returned for function a

c.#   ~ Call the function that was returned, directly!
d.#   ~ Call the function that was returned, directly!

Output: 

This is function b, inside of function a
This is function a, inside of function a

Since functions are referenced by name, we can do interesting tricks like defining an array of functions,
and call them all by name in a loop:

~ Declare an array of functions in 'a'

a{ "a!" }
b{ "b!" }
d{ "d!" }
e{ "e!" }
f{ "f!" }
c{ "c!" } ~ Note that these are defined out of order

0x: ( ax.+#   ~ Run the 'a'+x function
      x.1+x:  ~ Increment x
      x. 5 > ^  ~ If x > 5, exit loop
    )

Output:

a
b

c
d
e
f

Note that since we reference a variable by name, we are calling the functions by name, not directly 
order in which they were created.  However, this method would use up a lot of the 26 named variables. 
By nesting inside another function, we can regain the names back:

~ Function x prints out the English name for whatever
~ Value is passed into it

x { a { "Zero!" }
    f { "Five!" }
    b { "One!" }
    d { "Three!" }
    e { "Four!" }
    c { "Two!" }
    a+# } ~ Add the passed in value to the name a, and call
          ~ The function.  i.e. pass in 3, call a+3 (d)

0 a:     ~ a=0
( a. x#  ~ Call x's function #a
  a.1+a: ~ Increment a
  a. 5 > ^ ) ~ Loop until x > 5

And the output:

Zero
One
Two
Three
Four
Five

Memory Allocation

Finally, time for the final command, _ (underscore), the memory allocation operator.  Memory, or true 
anonymous variables (not referencable via offset), can be created by specifying the number of memory 
elements, followed by the _ (underscore) operator.  The address (not name) of the first variable that was
created is then put on the top of the stack.  For example, consider the next program:

~ Allocate one variable from memory

1_  ~ Allocate it
a:  ~ Save the address to 'a'

5 a.: ~ Assign the value 5 to the new memory (not a!!)

"Memory=" a..! "!"  ~ Print the value of the allocated memory
"And a=" a.! "!"    ~ Print the value of 'a'

The output:

Memory=5
And a=999

When we say “a.”, that means, take the name of the variable a, and push it to the stack.  Then, pop the 
name off the stack and replace it with the contents of a (999, in this case).  At that point, the top of the 
stack is holding the address of the memory we allocated.  We can then use another “.” operator to get 
the contents of the allocated memory.

Rather than using the predefined variables (a-z), even using additional offsets (like “a 100 +”), it is 
often cleaner to allocate a chunk of memory, even though it requires the use of the additional 
dereference operator (“.”) to avoid accidentally reusing an offset.  Here’s an example of where we 
allocate some memory, populate it, and print it out again.

~ Allocate a bunch of memory
5_a:  ~ Five memory addresses

1 a.: ~ Fill in the array
2 a.1+: ~ Fill in the array
3 a.2+: ~ Fill in the array
4 a.3+: ~ Fill in the array
5 a.4+: ~ Fill in the array

~ That was ugly; let's just use a loop to print it all out
0x:  ~ Initialize looping variable
(    ~ Start loop
    a.x.+.  ~ Get the variable of the allocated memory + x
    "Variable " x.! " = " ! "!"
    x.1+x:  ~ Increment x
    x. 4 >  ~ Is x > 4? (Array goes from 0-4, not 1-5)

    ^       ~ If so, exit
)

"Done.!"

That was a lot.  Here’s the output:

Variable 0 = 1
Variable 1 = 2
Variable 2 = 3
Variable 3 = 4
Variable 4 = 5
Done.

Note how, in the statement “a.3+:” when we fill in the array, we dereference the variable ‘a’ first before
adding the array element.  This is so we get the address of where we allocated memory before setting 
which position in the array we want to fill in.  If we did it in a different order, like “a3+.:”, this would 
mean, take the variable ‘d’ (a+3), get the data that d points, to, and assign the value on the top of the 
stack to that memory location.  This could be a useful thing to do, but not in this particular case.

Since functions can also be stored, here is the array of functions program, but using memory allocation:

~ Declare an array of functions in 'a'

6_a:

~ Assign the functions to the memory created

a.0+ { "a!" }
a.1+ { "b!" }
a.3+ { "d!" }
a.4+ { "e!" }
a.5+ { "f!" }
a.2+ { "c!" } ~ Note that these are defined out of order

0x: ( a.x.+.#   ~ Run the a+x function
      x.1+x:  ~ Increment x
      x. 5 > ^  ~ If x > 5, exit loop
    )

And the output:

a
b
c
d
e
f

In the statement “a.x.+.#”, what we are saying is: 
• Take the name ‘a’.  
• Change the name ‘a’ to what a points to (the memory allocated).
• Take x, and get the value of it. 
• Add the address of the memory allocated in variable ‘a’ to the value of x (the offset), 
• Get the value of that (the address of the function). 
• Call the function.  

It bears reiterating that the order of operations is highly important!  Not to confuse the issue, but “ax..
+” would mean “get the name a, then get the value of x, then push the value pointed to by the value of 
x, and add that to the name ‘a’.

Linked Lists

A linked list is a memory structure where one of the elements is a pointer to the next structure.  With 
this functionality, a whole chain of memory allocated structures can be connected together.  In Rev, we 
can create a linked list with the following:

~ Linked list functionality

~ Init a linked list. v=value; Returns a pointer to an element
i { v:         ~ Save off the initial value
    2_x:        ~ Allocate memory
    v.x.:       ~ Set the first value in memory to 'v'
    0 1 - x.1+: ~ Set the next value to -1
    x.          ~ Return the address of memory created
    "Created " v.! "!" }

~ Print out the linked list
p { x:         ~ x is a pointer to the first element
    "List: " 
    ( x..! " " ~ Print the element
      x.1+.x:  ~ Increment the pointer

      0 1- x. = ^ ~ If the next element is -1, exit loop
    ) "!" }       ~ Loop; print a new line

~ Add to the linked list
a { v: x:   ~ v = value, x = address of first element
    ~ skip to end of list
    ( x.1+. 0 1- = ^   ~ If the value of the next element
                       ~ is -1, break out of the loop
      x.1+.x: )        ~ x = the next pointer
    2_x.1+:   ~ Replace -1 with new memory
    x.1+.x:   ~ Set x to the new memory
    v.x.:     ~ First new memory value is the value
    0 1-x.1+: ~ Next new memory value if -1
    "Added " v.! "!" }

98 i# h: ~Initialize list to 98
h. p#
h. 99 a#  ~ Add 99 to the linked list
h. p#
h. 100 a# ~ Add 100
h. p#

And the output:

Created 98
List: 98 
Added 99
List: 98 99 
Added 100
List: 98 99 100 

Conclusion
Rev is an extremely flexible, powerful language, yet contains lots of pitfalls that memory allocation 
and pointers can often bring.  Regardless, it can be lots of fun to play with because of its dynamic 
capabilities and ability to represent coding concepts in a concise manner.

Reference
Symbol Description

~ Start of comment. Comment goes until end of line. 

(space) No action

$ End the program. Not required at the end of a program.

(number) Number. Push a number onto the stack. 

+ Add. Pop two operands from the stack and push their sum. 

- Subtract. Pop two operands from the stack and push their difference. 3 2 - results in 1. 

* Multiply. Pop two operands from the stack and push their product. 

/ Divide. Pop two operands from the stack and push their quotient. 9 3 / results in 3. 

% Remainder (modulus). Pop two operands from the stack and push the remainder of the 
first divided by the second. 

? Input number. Read a number from the keyboard and push it onto the stack. 

?’ Input character. Read a character from the keyboard and push its ASCII code onto the 
stack. 

! Output number. Pop a number from the stack and display it. 

!’ Output character. Pop an ASCII code from the stack and display the corresponding 
character on the terminal. 

“ Output string. Display each character between the " and the next " on the terminal. 
Translate the character ! to a new line. 

‘(char) Character literal. Push the ASCII code corresponding the the character onto the stack. 

(letter) Push the character equivalent (a=0, b=1...) onto the stack as a name.

: Assignment. Pop two operands from the stack and store the value of the second at the 
address specified by the first. To assign 3 to x, use "3 x:"

. Dereference. If the value on the top of the stack is a name, return the value of the name. 
If it is an address, push the value stored at that address onto the stack. 

< Comparison. Pop two operands from the stack. If the first is less than the second, push 
1, else push 0 onto the stack. 

= Comparison. Pop two operands from the stack. If they are equal, push 1, else push 0 
onto the stack. 

> Comparison. Pop two operands from the stack. If the first is greater than the second, 
push 1, else push 0 onto the stack. 

[ Conditional. Pop a value from the stack. If it is zero, skip until a matching ] is found. 

] Matches opening bracket 

( Start loop. No action.  Matches closing parenthesis. 

) Go back to the beginning of the associated loop. 

^ Pops a value; if it is not zero, exit loop. 

# Pop a value, then call a function by the specified name. To call function 'a', use 'a#'. 

@ Return early from a function 

{ Start of function.  Pops a value off the stack, and sets the value of that variable to the 
address of the function.  To declare a function ‘a’, use ‘a{ … }’

} End of function

_ Memory allocation.  Pops a value off the stack, allocates that many values, and returns a
pointer to the first value.

